
: EMBEDCOMP ( xt--- ) \ Compiler for children of EMBED
  >BODY @ POSTPONE LITERAL
;

: EMBED { pzpath | pfile plen ppos -- } \ Embed a file within the Forth directory
  pzpath R/O OPEN-ZFILE SWAP -> pfile 0= IF \ File opened OK
    pfile FILE-SIZE -ROT D>S -> plen 0= plen 0<> AND IF \ File size ok and is non-zero
      plen CELL+ IRESERVE -> ppos \ Get address of reserved space
      plen ppos ! \ Save length
      ppos CELL+ plen pfile READ-FILE 0= SWAP plen = AND 0= IF \ Read in OK
        CR Z" Cannot read file " pzpath Z+ Z"  to embed" Z+ Z$. ABORT
      THEN
    ELSE \ No file size
      CR Z" Failed to embed empty file " pzpath Z+ Z+ Z$. ABORT
    THEN
    pfile CLOSE-FILE DROP \ Close file
  ELSE \ Cannot open file
    CR Z" Cannot open file " pzpath Z+ Z"  to embed" Z+ Z$. ABORT
  THEN
  CR Z" Embedding " pzpath z+ z$.
  pzpath ZCOUNT ($CREATE) \ Create Forth word from file path
    ppos , ['] EMBEDCOMP set-compiler \ Returns the address of file length
  interp>
    @
;

4 1 callproc: EMBEDDIRFUN { ppath pstat ptype pbuf -- n } \ Callback function from walk dir tree
  ptype 0= IF \ Only files
    ppath ZCOUNT 1- + C@ [CHAR] ~ <> IF \ No Glade backup files
      ppath EMBED \ Embed file
    THEN
  THEN
  FALSE \ Continue
;

: EMBEDDIR { pzdir | pdir -- } \ Embed all files in directory
  pzdir EMBEDDIRFUN 0 0 nftw IF \ Walk directory
    CR Z" Failed to embed files from directory " pzdir Z+ Z+ Z$. ABORT
  THEN
;

SVGICONSDIR   EMBEDDIR \ Embed SVG icon files

: DEBEDICON { picon | pbed pdest pfile -- } \ Copy embedded svg icon to the tracknet temp directory
  SVGICONSDIR picon Z+ ZCOUNT SEARCH-CONTEXT IF \ Embedded file found
    EXECUTE -> pbed \ Set embedded address
    TEMPDIR picon Z+ -> pdest \ Construct destination path
    pdest ZCOUNT DELETE-FILE DROP \ Delete any previous file
    pdest R/W CREATE-ZFILE SWAP -> pfile 0= IF \ Temporary file opened OK
      pbed CELL+ pbed @ pfile WRITE-FILE DROP \ Write image to temporary file
      pfile CLOSE-FILE DROP
    ELSE \ Failed to create file
      Z" Failed to create debedded file " picon Z+ FATAL
    THEN
  ELSE \ Failed to find embedded file
    Z" Failed to find file " picon Z+ Z"  to debed" Z+ FATAL
  THEN
;

...
  Z" caret-right-solid.svg" DEBEDICON
...

: LOAD-SCALED-PIXBUF? { zpath | paddr pfile -- ppb } \ Get scaled pixbuf from embedded, false if fail
  zpath ZCOUNT SEARCH-CONTEXT IF \ Embedded file found
    EXECUTE -> paddr \ Get address of count
    PIXBUF-TEMPFILE ZCOUNT DELETE-FILE DROP \ Delete any previous file
    PIXBUF-TEMPFILE R/W CREATE-ZFILE SWAP -> pfile 0= IF \ Temporary file opened OK
      paddr CELL+ paddr @ pfile WRITE-FILE DROP \ Write image to temporary file
      pfile CLOSE-FILE DROP
      PIXBUF-TEMPFILE 16 REMSCALE DUP \ Scale
      TRUE NULL gdk_pixbuf_new_from_file_at_scale
    ELSE
      FALSE
    THEN
  ELSE
    FALSE
  THEN
;

